

---

COMPUTER SCIENCE MENTORS 61AOctober 31–November 4, 2022

---

---

1 Scheme

---

Scheme is a *functional* language, as opposed to Python, which is an *imperative* language. A Python program is comprised of *statements* or “instructions,” each of which directs the computer to take some action. (An example of a statement would be something like `x = 3` in Python. This does not evaluate to any value but just instructs Python to create a variable `x` with the value 3.) In contrast, a Scheme program is composed solely of (often heavily nested) *expressions*, each of which simply evaluates to a value.

The three basic types of expressions in Scheme are atomics/primitive expressions, call expressions, and special forms.

---

1.1 Atomics

---

Atomics are the simplest expressions. Some atomics, such as numbers and booleans, are called “self-evaluating” because they evaluate to themselves:

- `123`  $\rightarrow$  `123`
- `-3.14`  $\rightarrow$  `-3.14`
- `#t`  $\rightarrow$  `#t` ;booleans in scheme are `#t` and `#f`

**Symbols** (variables) are also atomic expressions; they evaluate to the values to which they are bound. For example, the symbol `+` evaluates to the addition **procedure** (function) `# [+]`.

---

1.2 Call Expressions

---

A call expression is denoted with parentheses and is formed like so:

`(<operator> <operand0> <operand1> ... <operand>)`

Each “element” of a call expression is an expression itself and is separated from its neighbors by whitespace. All call expressions are evaluated in the same way:

1. Evaluate the operator, which will return a procedure.
2. Evaluate the operands.
3. Apply operator on operands.

For example, to evaluate the call expression `(+ (+ 1 2) 2)`, we first evaluate the operator `+`, which returns the procedure `#[+]`. Then we evaluate the first operand `(+ 1 2)`, which returns 3. Then, we evaluate the last operand 2, receiving 2. Finally, we apply the `#[+]` procedure to 3 and 2, which returns 5. So this call expression evaluates to 5.

Note that in order to add two numbers, we had to call a function. In Python, `+` is a binary operator that can add two numbers without calling a function. Scheme has no such constructs, so even the most basic arithmetic requires you to call a function. The other math operators, including `-` (both subtraction and negation), `*`, `/`, `expt` (exponentiation), `=`, `<`, `>`, `<=`, and `>=` function in the same way.

### 1.3 Special Forms

---

Special forms *look* just like call expressions but are distinct in two ways:

1. One of the following keywords appears in the operator slot: **define**, **if**, **cond**, **and**, **or**, **let**, **begin**, **lambda**, **quote**, **quasiquote**, **unquote**, **mu**, **define-macro**, **expect**, **unquote-splicing**, **delay**, **cons-stream**, **set!**
2. They do not follow the evaluation rules for call expressions.

Below, we will go through a few commonly seen special forms.

#### 1.3.1 **if** expression

```
(if <predicate> <true-expr> <false-expr>)
```

An **if** expression is similar to a Python **if** statement. First, evaluate `<predicate>`.

- If `<predicate>` is true, evaluate and return `<true-expr>`.
- If `<predicate>` is false, evaluate and return `<false-expr>`.

Note that everything in Scheme is truthy (including 0) except for `#f`.

Also note that in Python, **if** is a statement, whereas in Scheme, **if** is an expression that evaluates to a value like any other expression would. In Scheme, you can then write something like this:

```
scm> (+ 1 (if #t 9 99))  
10
```

Other special forms are also expressions that evaluate to values. Therefore, when we say “returns  $x$ ,” we mean “the special form evaluates to  $x$ .”

### 1.3.2 **cond** expression

```
(cond
  (<predicate1> <expr1>)
  ...
  (<predicateN> <exprN>)
  (else <else-expr>))
```

A **cond** expression is similar to a Python **if-elif-else** statement. It is an alternative to using many nested **if** expressions.

- Evaluate <predicate1>. If it is true, evaluate and return <expr1>.
- Otherwise, continue down the list by evaluating <predicate2>. If it is true, evaluate and return <expr2>.
- Continue in this fashion down the list until you hit a true predicate.
- If every predicate is false, return <else-expr>.

### 1.3.3 **and** expression

```
(and <expr1> ... <exprN>)
```

**and** in Scheme works similarly to **and** in Python. Evaluate the expressions in order and return the value of the first false expression. If all of the values are true, return the last value. If no operands are provided, return #t.

### 1.3.4 **or** expression

```
(or <expr1> ... <exprN>)
```

**or** in Scheme works similarly to **or** in Python. Evaluate the expressions in order and return the value of the first true expression. If all of the values are false, return the last value. If no operands are provided, return #f.

### 1.3.5 **define** expression

**define** does two things. It can define variables, similar to the Python = assignment operator:

```
(define <symbol> <expr>)
```

This will evaluate <expr> and bind the resulting value to <symbol> in the current frame. **define** is also used to define procedures.

```
(define (<symbol> <op1> ... <opN>)
  <body>)
```

This code will create a new procedure that takes in the formal parameters <op1> ... <opN> and bind it to <symbol> in the current frame. When that procedure is called, the <body>, which may have multiple expressions, will be executed with the provided arguments bound to <op1> ... <opN>. The value of the final expression of <body> will be returned.

With either version of **define**, <symbol> is returned.

Dealing with the different types of **define** can be tricky. Scheme differentiates between the two by whether the first operand is a symbol or a list:

```
(define (x) 1) ; like x = lambda: 1
(define x 1) ; like x = 1
```

### 1.3.6 lambda expressions

```
(lambda (<op1> ... <opN>)
  <body>)
```

Returns a new procedure that takes in the formal parameters <op1> ... <opN>. When that procedure is called, the <body>, which may have multiple expressions, will be executed with the provided arguments bound to <op1> ... <opN>. The value of the final expression of <body> will be returned.

### 1.3.7 begin special form

```
(begin
  <expr1>
  ...
  <exprN>)
```

Evaluates <expr1>, <expr2>, ..., <exprN> in order in the current environment. Returns the value of <exprN>.

### 1.3.8 let special form

```
(let ((<symbol1> <expr1>)
  ...
  (<symbolN> <exprN>))
  <body>)
```

Evaluates <expr1>, ..., <exprN> in the current environment. Then, creates a new frame as a child of the current frame and binds the values of <expr1>, ..., <exprN> to <symbol1>,

..., <symbolN>, respectively, in that new frame. Finally, Scheme evaluates the <body>, which may have multiple expressions, in the new frame. The value of the final expression of <body> is returned.

### 1.3.9 quote special form

(**quote** <expr>)  
'<expr> ; shorthand syntax

Returns an expression that evaluates to <expr> *in its unevaluated form*. In other words, if you put '<expr> into the Scheme interpreter, you should get <expr> out *exactly*.

### 1.3.10 Summary of special forms

We have presented the main details of the most important special forms here, but this account is not comprehensive. Please see <https://cs61a.org/articles/scheme-spec/> for a fuller explanation of the Scheme language.

behavior	syntax
if/else	(if <predicate> <true-expr> <false-expr>)
if/elif/else	( <b>cond</b> (<predicate1> <expr1>) ... (<predicateN> <exprN>) ( <b>else</b> <else-expr>))
and	(and <expr1> ... <exprN>)
or	(or <expr1> ... <exprN>)
variable assignment	(define <symbol> <expr>)
function definition	( <b>define</b> (<symbol> <op1> ... <opN>) <body>)
lambdas	( <b>lambda</b> (<op1> ... <opN>) <body>)
evaluate many lines	( <b>begin</b> <expr1> ... <exprN>)
temporary environment	( <b>let</b> ((<symbol1> <expr1>) ... (<symbolN> <exprN>)) <body>)
quote	(quote <expr>) or '<expr>

## 1. What will Scheme output?

```
scm> (define pi 3.14)
```

```
pi
```

```
scm> pi
```

```
3.14
```

```
scm> 'pi
```

```
pi
```

```
scm> (+ 1 2)
```

```
3
```

```
scm> (+ 1 (* 3 4))
```

```
13
```

```
scm> (if 2 3 4)
```

```
3
```

```
scm> (if 0 3 4)
```

```
3
```

```
scm> (- 5 (if #f 3 4))
```

```
1
```

```
scm> (if (= 1 1) 'hello 'goodbye)
```

```
hello
```

```
scm> (define (factorial n)
      (if (= n 0)
          1
          (* n (factorial (- n 1)))))
```

```
factorial
```

```
scm> (factorial 5)
```

```
120
```

2. Define `apply-multiple` which takes in a single argument function `f`, a nonnegative integer `n`, and a value `x` and returns the result of applying `f` to `x` a total of `n` times.

```
;doctests
scm> (apply-multiple (lambda (x) (* x x)) 3 2)
256
scm> (apply-multiple (lambda (x) (+ x 1)) 10 1)
11
scm> (apply-multiple (lambda (x) (* 1000 x)) 0 5)
5
```

```
(define (apply-multiple f n x)
```

```
)
```

```
(define (apply-multiple f n x)
  (if (= n 0)
      x
      (f (apply-multiple f (- n 1) x)))))
```

Alternate solution:

```
(define (apply-multiple f n x)
  (if (= n 0)
      x
      (apply-multiple f (- n 1) (f x)))))
```

3. Define a procedure called `hailstone`, which takes in two numbers `seed` and `n` and returns the `n`th number in the hailstone sequence starting at `seed`. Assume the hailstone sequence starting at `seed` has a length of at least `n`. As a reminder, to get the next number in the sequence, divide by 2 if the current number is even. Otherwise, multiply by 3 and add 1.

#### Useful procedures

- `quotient`: floor divides, much like `//` in python  
`(quotient 103 10)` outputs 10
- `remainder`: takes two numbers and computes the remainder of dividing the first number by the second  
`(remainder 103 10)` outputs 3

```
; The hailstone sequence starting at seed = 10 would be  
; 10 => 5 => 16 => 8 => 4 => 2 => 1
```

```
; Doctests  
> (hailstone 10 0)  
10  
> (hailstone 10 1)  
5  
> (hailstone 10 2)  
16  
> (hailstone 5 1)  
16
```

```
(define (hailstone seed n)
```

```
)
```



```

(define (hailstone seed n)
  (if (= n 0)
      seed
      (if (= 0 (remainder seed 2))
          (hailstone
            (quotient seed 2)
            (- n 1))
          (hailstone
            (+ 1 (* seed 3))
            (- n 1))))))

```

; Alternative solution with cond

```

(define (hailstone seed n)
  (cond
    ((= n 0) seed)
    ((= 0 (remainder seed 2))
     (hailstone
      (quotient seed 2)
      (- n 1)))
    (else
     (hailstone
      (+ 1 (* seed 3))
      (- n 1)))))

```

## 2 Scheme Lists

Unlike Python, all Scheme lists are linked lists. Recall that, in Python, a linked list is made up of `Link`s that each have a `first` and a `rest`, where the `rest` is another `Link`. Similarly, each Scheme list is a “pair” where the first element of the pair is the first element of the list, and the second element of the pair is the rest of the list (also a pair).

We use the `cons` procedure to construct Scheme lists, and `nil` to represent empty lists. The sequence 1, 2, 3 may then be represented as follows:

```

scm> (cons 1 (cons 2 (cons 3 nil)))
(1 2 3)

```

The `car` and `cdr` procedures are used to access the elements of a Scheme list. `car` gets the first element of a list, while `cdr` gets the rest of the list:

```

scm> (define lst (cons 1 (cons 2 (cons 3 nil))))

```

```

lst
scm> (car lst)
1
scm> (cdr lst)
(2 3)

```

You can make the following analogy between linked lists in Python and Scheme:

<pre> Link(1, Link.empty) a = Link(1, Link(2, Link.empty)) a.first a.rest </pre>	<pre> (cons 1 nil) (define a (cons 1 (cons 2 nil))) (car a) (cdr a) </pre>
--	--

The `list` procedure and quotation give us additional convenient ways to construct lists:

```

scm> (list 1 2 3)
(1 2 3)
scm> '(1 2 3)
(1 2 3)
scm> (list 1 (+ 1 1) 3)
(1 2 3)
scm> '(1 (+ 1 1) 3)
(1 (+ 1 1) 3)

```

Note that quotation will prevent any of the list items from being evaluated, which can occasionally be inconvenient.

## 2.1 Useful procedures

In addition to the procedures mentioned above, the following procedures are often useful when dealing with Scheme lists:

- `(null? s)`: returns true if `s` is `nil`.
- `(length s)`: returns the length of `s`.
- `(append s1 ... sn)`: returns the result of concatenating lists `s1, ..., sn`.
- `(map f s)`: returns the result of applying the procedure `f` to each element of `s`.
- `(filter pred s)`: returns a list containing the elements of `s` for which the single-argument procedure `pred` returns true.
- `(reduce comb s)`: combines the elements of `s` into a single value using the two-argument procedure `comb`.

## 2.2 Equality testing

---


Equality testing in Scheme is a bit confusing as it is handled by three separate procedures:

- `(= a b)`: returns true if `a` equals `b`. Both must be numbers.
- `(eq? a b)`: returns true if `a` and `b` are equivalent primitive values. For two objects, `eq?` returns true if both refer to the exactly same object in memory (like `is` in Python).
- `(equal? a b)`: returns true if `a` and `b` are equivalent. Two lists are equivalent if their elements are equivalent.

1. What will Scheme output? Draw box-and-pointer diagrams to help determine this. (Ask your mentor if you're unsure what's going on. You aren't expected to understand this completely on your own.)

```
scm> (cons 1 (cons 2 nil))
```

```
(1 2)
```



```
scm> (cons 1 '(2 3 4 5))
```


```
(1 2 3 4 5)
```



When we use the quote before the list, we are saying that we should put the literal list (2 3 4 5) in the cdr of this list. So in this case we create a list where the first element (car) is 1, and the cdr is the list (2 3 4 5).

```
scm> (cons 1 '(2 (cons 3 nil)))
```

```
(1 2 (cons 3 ()))
```



Since we also used a quote here, we do not evaluate the (cons 3 nil). We keep everything inside the quotes the same so the cdr of this list is the list (2 (cons 3 nil)). That means that we add the element 2, and then the nested list (cons 3 nil).

```
scm> (cons 1 (2 (cons 3 nil)))
```

```
eval: bad function in : (2 (cons 3 nil))
```

While evaluating the operands, Scheme will try to evaluate the expression (2 (cons 3 nil)). Since 2 is not a valid operator, this expression Errors.

```
scm> (cons 3 (cons (cons 4 nil) nil))
```

```
(3 (4))
```


```
scm> (define a '(1 2 3))
```

```
a
```

Defines a list of elements of (1 2 3) and binds the list to the variable a. Recall that define returns the name of the symbol.

```
scm> a
```

```
(1 2 3)
```



```
scm> (car a)
```

```
1
```

```
scm> (cdr a)
```

```
(2 3)
```

```
scm> (car (cdr a))
```

```
2
```

From above, we know that (cdr a) is (2 3). From that, we can evaluate (car (cdr a)) to 2.

How can we get the 3 out of a?

```
(car (cdr (cdr a)))
```

To get to the pair that contains 3, we need to call (cdr (cdr a)). To get the element 3, we need the car of (cdr (cdr a)).

2. Define `is-prefix`, which takes in a list `p` and a list `lst` and determines if `p` is a prefix of `lst`. That is, it determines if `lst` starts with all the elements in `p`.

```
; Doctests:  
scm> (is-prefix '() '())  
#t  
scm> (is-prefix '() '(1 2))  
#t  
scm> (is-prefix '(1) '(1 2))  
#t  
scm> (is-prefix '(2) '(1 2))  
#f  
; Note here p is longer than lst  
scm> (is-prefix '(1 2) '(1))  
#f
```

```
(define (is-prefix p lst)
```

```
)
```

```

; is-prefix with nested if statements
(define (is-prefix p lst)
  (if (null? p)
      #t
      (if (null? lst)
          #f
          (and
            (= (car p) (car lst))
            (is-prefix (cdr p) (cdr lst))))))

; is-prefix with a cond statement
(define (is-prefix p lst)
  (cond
    ((null? p) #t)
    ((null? lst) #f)
    (else (and (= (car p) (car lst))
                 (is-prefix (cdr p) (cdr lst))))))

```

3. Implement `argmax`, a function that takes in a list, `lst`, and returns the index of the largest element in `lst`. If there are two or more elements that are the largest element, return the index of the one that appears first in `lst`.

You can assume all elements of `lst` are non-negative integers, and `lst` has at least 1 element and no nested lists.

```

(define (argmax lst)
  (define (max-helper lst max-so-far max-index curr-index)
    (cond
      (( (_____)) _____)
      (( (_____)) _____
        _____)
      (else _____)
    )
  )
  (max-helper _____)

```

```

)

(define (argmax lst)
  (define (max-helper lst max-so-far max-index curr-index)
    (cond
      ((null? lst) max-index)
      ((> (car lst) max-so-far)
       (max-helper (cdr lst) (car lst) curr-index (+
        curr-index 1)))
      (else
       (max-helper (cdr lst) max-so-far max-index (+
        curr-index 1))))
    )
  )
  (max-helper lst 0 0 0)
)

```